

Cardinality of Multisets

Lukas Bulwahn

June 17, 2024

Abstract

This entry provides three lemmas to count the number of multisets of a given size and finite carrier set. The first lemma provides a cardinality formula assuming that the multiset's elements are chosen from the given carrier set. The latter two lemmas provide formulas assuming that the multiset's elements also cover the given carrier set, i.e., each element of the carrier set occurs in the multiset at least once.

The proof of the first lemma uses the argument of the recurrence relation for counting multisets [1]. The proof of the second lemma is straightforward, and the proof of the third lemma is easily obtained using the first cardinality lemma. A challenge for the formalization is the derivation of the required induction rule, which is a special combination of the induction rules for finite sets and natural numbers. The induction rule is derived by defining a suitable inductive predicate and transforming the predicate's induction rule.

Contents

1	Cardinality of Multisets	1
1.1	Additions to Multiset Theory	1
1.2	Lemma to Enumerate Sets of Multisets	2
1.3	Derivation of Suitable Induction Rule	3
1.4	Finiteness of Sets of Multisets	4
1.5	Cardinality of Multisets	4

1 Cardinality of Multisets

```
theory Card-Multisets
imports
  HOL-Library.Multiset
begin
```

1.1 Additions to Multiset Theory

```
lemma mset-set-set-mset-subseteq:
```

```

  mset-set (set-mset M)  $\subseteq$  # M
proof (induct M)
  case empty
  show ?case by simp
next
  case (add x M)
  from this show ?case
proof (cases x  $\in$  # M)
  assume x  $\in$  # M
  from this have mset-set (set-mset (M + {#x#})) = mset-set (set-mset M)
    by (simp add: insert-absorb)
  from this add.hyps show ?thesis
    using subset-mset.trans by fastforce
next
  assume  $\neg$  x  $\in$  # M
  from this add.hyps have {#x#} + mset-set (set-mset M)  $\subseteq$  # M + {#x#}
    by (simp add: insert-subset-eq-iff)
  from this  $\langle \neg$  x  $\in$  # M  $\rangle$  show ?thesis by simp
qed
qed

```

```

lemma size-mset-set-eq-card:
  assumes finite A
  shows size (mset-set A) = card A
using assms by (induct A) auto

```

```

lemma card-set-mset-leq:
  card (set-mset M)  $\leq$  size M
by (induct M) (auto simp add: card-insert-le-m1)

```

1.2 Lemma to Enumerate Sets of Multisets

```

lemma set-of-multisets-eq:
  assumes x  $\notin$  A
  shows {M. set-mset M  $\subseteq$  insert x A  $\wedge$  size M = Suc k} =
    {M. set-mset M  $\subseteq$  A  $\wedge$  size M = Suc k}  $\cup$ 
    ( $\lambda$ M. M + {#x#}) ‘ {M. set-mset M  $\subseteq$  insert x A  $\wedge$  size M = k}
proof –
  from  $\langle$  x  $\notin$  A  $\rangle$  have {M. set-mset M  $\subseteq$  insert x A  $\wedge$  size M = Suc k} =
    {M. set-mset M  $\subseteq$  A  $\wedge$  size M = Suc k}  $\cup$ 
    {M. set-mset M  $\subseteq$  insert x A  $\wedge$  size M = Suc k  $\wedge$  x  $\in$  # M}
  by auto
  moreover have {M. set-mset M  $\subseteq$  insert x A  $\wedge$  size M = Suc k  $\wedge$  x  $\in$  # M} =
    ( $\lambda$ M. M + {#x#}) ‘ {M. set-mset M  $\subseteq$  insert x A  $\wedge$  size M = k} (is ?S =
    ?T)
proof
  show ?S  $\subseteq$  ?T
proof
  fix M

```

```

assume  $M \in ?S$ 
from this have  $M = M - \{\#x\# \} + \{\#x\# \}$  by auto
moreover have  $M - \{\#x\# \} \in \{M. \text{set-mset } M \subseteq \text{insert } x \ A \wedge \text{size } M = k\}$ 
proof –
  have  $\text{set-mset } (M - \{\#x\# \} + \{\#x\# \}) \subseteq \text{insert } x \ A$ 
    using  $\langle M \in ?S \rangle$  by force
    moreover have  $\text{size } (M - \{\#x\# \} + \{\#x\# \}) = \text{Suc } k \wedge x \in \# \ M -$ 
 $\{\#x\# \} + \{\#x\# \}$ 
    using  $\langle M \in ?S \rangle$  by force
    ultimately show ?thesis by force
  qed
  ultimately show  $M \in ?T$  by auto
qed
next
  show  $?T \subseteq ?S$  by force
qed
ultimately show ?thesis by auto
qed

```

1.3 Derivation of Suitable Induction Rule

context

begin

private inductive $R :: 'a \text{ set} \Rightarrow \text{nat} \Rightarrow \text{bool}$

where

$\text{finite } A \Longrightarrow R \ A \ 0$

$| R \ \{\} \ k$

$| \text{finite } A \Longrightarrow x \notin A \Longrightarrow R \ A \ (\text{Suc } k) \Longrightarrow R \ (\text{insert } x \ A) \ k \Longrightarrow R \ (\text{insert } x \ A)$
 $(\text{Suc } k)$

private lemma *R-eq-finite:*

$R \ A \ k \longleftrightarrow \text{finite } A$

proof

assume $R \ A \ k$

from this show *finite A* **by** *cases auto*

next

assume *finite A*

from this show $R \ A \ k$

proof (*induct A*)

case *empty*

from this show *?case* **by** (*rule R.intros(2)*)

next

case *insert*

from this show *?case*

proof (*induct k*)

case *0*

from this show *?case*

by (*intro R.intros(1) finite.insertI*)

```

next
  case Suc
  from this show ?case
    by (metis R.simps Zero-neq-Suc diff-Suc-1)
qed
qed
qed

```

lemma *finite-set-and-nat-induct*[*consumes 1, case-names zero empty step*]:

```

assumes finite A
assumes  $\bigwedge A. \text{finite } A \implies P A 0$ 
assumes  $\bigwedge k. P \{ \} k$ 
assumes  $\bigwedge A k x. \text{finite } A \implies x \notin A \implies P A (Suc k) \implies P (\text{insert } x A) k \implies$ 
 $P (\text{insert } x A) (Suc k)$ 
shows  $P A k$ 
proof -
  from  $\langle \text{finite } A \rangle$  have  $R A k$  by (subst R-eq-finite)
  from this assms(2-4) show ?thesis by (induct A k) auto
qed

```

end

1.4 Finiteness of Sets of Multisets

lemma *finite-multisets*:

```

assumes finite A
shows finite  $\{M. \text{set-mset } M \subseteq A \wedge \text{size } M = k\}$ 
using assms
proof (induct A k rule: finite-set-and-nat-induct)
  case zero
  from this show ?case by auto
next
  case empty
  from this show ?case by auto
next
  case (step A k x)
  from this show ?case
    using set-of-multisets-eq[OF  $\langle x \notin A \rangle$ ] by simp
qed

```

1.5 Cardinality of Multisets

lemma *card-multisets*:

```

assumes finite A
shows  $\text{card } \{M. \text{set-mset } M \subseteq A \wedge \text{size } M = k\} = (\text{card } A + k - 1) \text{ choose } k$ 
using assms
proof (induct A k rule: finite-set-and-nat-induct)
  case (zero A)
  assume finite ( $A :: 'a \text{ set}$ )
  have  $\{M. \text{set-mset } M \subseteq A \wedge \text{size } M = 0\} = \{\{\#\}\}$  by auto

```

from this show $\text{card } \{M. \text{set-mset } M \subseteq A \wedge \text{size } M = 0\} = \text{card } A + 0 - 1$
choose 0
by simp
next
case (*empty k*)
show $\text{card } \{M. \text{set-mset } M \subseteq \{\} \wedge \text{size } M = k\} = \text{card } \{\} + k - 1$ *choose k*
by (*cases k*) (*auto simp add: binomial-eq-0*)
next
case (*step A k x*)
let $?S_1 = \{M. \text{set-mset } M \subseteq A \wedge \text{size } M = \text{Suc } k\}$
and $?S_2 = \{M. \text{set-mset } M \subseteq \text{insert } x \ A \wedge \text{size } M = k\}$
assume *hyp1*: $\text{card } ?S_1 = \text{card } A + \text{Suc } k - 1$ *choose Suc k*
assume *hyp2*: $\text{card } ?S_2 = \text{card } (\text{insert } x \ A) + k - 1$ *choose k*
have *finite-sets*: *finite* $?S_1$ *finite* $((\lambda M. M + \{\#x\#}) \ ` ?S_2)$
using $\langle \text{finite } A \rangle$ **by** (*auto simp add: finite-multisets*)
have *inj*: *inj-on* $(\lambda M. M + \{\#x\#}) \ ?S_2$ **by** (*rule inj-onI*) *auto*
have $\text{card } \{M. \text{set-mset } M \subseteq \text{insert } x \ A \wedge \text{size } M = \text{Suc } k\} =$
 $\text{card } (?S_1 \cup (\lambda M. M + \{\#x\#}) \ ` ?S_2)$
using *set-of-multisets-eq* $\langle x \notin A \rangle$ **by** *fastforce*
also have $\dots = \text{card } ?S_1 + \text{card } ((\lambda M. M + \{\#x\#}) \ ` ?S_2)$
using *finite-sets* $\langle x \notin A \rangle$ **by** (*subst card-Un-disjoint*) *auto*
also have $\dots = \text{card } ?S_1 + \text{card } ?S_2$
using *inj* **by** (*auto intro: card-image*)
also have $\dots = (\text{card } A + \text{Suc } k - 1 \text{ choose } \text{Suc } k) + (\text{card } (\text{insert } x \ A) + k -$
 $1 \text{ choose } k)$
using *hyp1 hyp2* **by** *simp*
also have $\dots = \text{card } (\text{insert } x \ A) + \text{Suc } k - 1$ *choose Suc k*
using $\langle x \notin A \rangle \langle \text{finite } A \rangle$ **by** *simp*
finally show *?case* .
qed

lemma *card-too-small-multisets-covering-set*:

assumes *finite A*
assumes $k < \text{card } A$
shows $\text{card } \{M. \text{set-mset } M = A \wedge \text{size } M = k\} = 0$
proof –
from $\langle k < \text{card } A \rangle$ **have** *eq*: $\{M. \text{set-mset } M = A \wedge \text{size } M = k\} = \{\}$
using *card-set-mset-leq Collect-empty-eq leD* **by** *auto*
from this show *?thesis* **by** (*metis card.empty*)
qed

lemma *card-multisets-covering-set*:

assumes *finite A*
assumes $\text{card } A \leq k$
shows $\text{card } \{M. \text{set-mset } M = A \wedge \text{size } M = k\} = (k - 1) \text{ choose } (k - \text{card } A)$
proof –
have $\{M. \text{set-mset } M = A \wedge \text{size } M = k\} = (\lambda M. M + \text{mset-set } A) \ `$
 $\{M. \text{set-mset } M \subseteq A \wedge \text{size } M = k - \text{card } A\}$ (**is** $?S = ?f \ ` ?T$)
proof

```

show  $?S \subseteq ?f \text{ ' } ?T$ 
proof
  fix  $M$ 
  assume  $M \in ?S$ 
  from this have  $M = M - \text{mset-set } A + \text{mset-set } A$ 
    by (auto simp add: mset-set-set-mset-subseteq subset-mset.diff-add)
  moreover from  $\langle M \in ?S \rangle$  have  $M - \text{mset-set } A \in ?T$ 
    by (auto simp add: mset-set-set-mset-subseteq size-Diff-submset size-mset-set-eq-card
in-diffD)
  ultimately show  $M \in ?f \text{ ' } ?T$  by auto
qed
next
  from  $\langle \text{finite } A \rangle \langle \text{card } A \leq k \rangle$  show  $?f \text{ ' } ?T \subseteq ?S$ 
    by (auto simp add: size-mset-set-eq-card)+
qed
  moreover have inj-on  $?f \text{ ' } ?T$  by (rule inj-onI) auto
  ultimately have  $\text{card } ?S = \text{card } ?T$  by (simp add: card-image)
  also have  $\dots = \text{card } A + (k - \text{card } A) - 1$  choose  $(k - \text{card } A)$ 
    using  $\langle \text{finite } A \rangle$  by (simp only: card-multisets)
  also have  $\dots = (k - 1)$  choose  $(k - \text{card } A)$ 
    using  $\langle \text{card } A \leq k \rangle$  by auto
  finally show ?thesis .
qed
end

```

References

- [1] Wikipedia. Multiset — wikipedia, the free encyclopedia, 2016. [Online; accessed 23-June-2016].